// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/interfaces/IUniswapV2Router01.sol



// SPDX-License-Identifier: MIT
pragma solidity =0.8.16;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);
    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);
    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);
    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);
    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountToken, uint amountETH);
    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    function swapTokensForExactETH(uint amountOut, uint amountInMax, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapETHForExactTokens(uint amountOut, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function quote(uint amountA, uint reserveA, uint reserveB) external pure returns (uint amountB);
    function getAmountOut(uint amountIn, uint reserveIn, uint reserveOut) external pure returns (uint amountOut);
    function getAmountIn(uint amountOut, uint reserveIn, uint reserveOut) external pure returns (uint amountIn);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}

// File: contracts/UniswapV2Router01IntermediaryV2.sol



// ███╗░░░███╗████████╗░█████╗░██████╗░░██████╗░██╗░░░░░░░██╗░█████╗░██████╗░
// ████╗░████║╚══██╔══╝██╔══██╗██╔══██╗██╔════╝░██║░░██╗░░██║██╔══██╗██╔══██╗
// ██╔████╔██║░░░██║░░░██║░░██║██████╔╝╚█████╗░░╚██╗████╗██╔╝███████║██████╔╝
// ██║╚██╔╝██║░░░██║░░░██║░░██║██╔═══╝░░╚═══██╗░░████╔═████║░██╔══██║██╔═══╝░
// ██║░╚═╝░██║░░░██║░░░╚█████╔╝██║░░░░░██████╔╝░░╚██╔╝░╚██╔╝░██║░░██║██║░░░░░
// ╚═╝░░░░░╚═╝░░░╚═╝░░░░╚════╝░╚═╝░░░░░╚═════╝░░░░╚═╝░░░╚═╝░░╚═╝░░╚═╝╚═╝░░░░░

// SPDX-License-Identifier: MIT
pragma solidity =0.8.16;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";

import "./interfaces/IUniswapV2Router01.sol";

/**
 * @dev UniswapV2Router01IntermediaryV2 is an interface between the 'real' UniswapV2Router01
 * and MtopSwap's Dapp. This allows differentiating the transactions that come from MtopSwap
 * or other services. This is useful for e.g. DappRadar.
 *
 * The fee functions are intended to be used when a user hasn't subscribed to MtopSwap, but
 * already used the free trial.
 */
contract UniswapV2Router01IntermediaryV2 is Ownable {
    using SafeERC20 for IERC20;

    /***************
     ** Variables **
     ***************/

    address public feeReceiver;
    uint16 public feeBP = 25; // = 0.25 %

    /***************
     ** Constants **
     ***************/

    /**
     * @dev BP = Percent * 100
     * @notice BP has to be <= FEE_DENOMINATOR
     * BP = FEE_DENOMINATOR -> 100 %
     */
    uint16 public constant FEE_DENOMINATOR = 10000;
    uint16 public constant MAX_FEE_BP = 100; // = 1 %

    /*****************
     ** Constructor **
     *****************/

    constructor(address feeReceiver_) {
        _setFeeReceiver(feeReceiver_);
    }

    /************
     ** Events **
     ************/

    event FeeBPChange(uint16 indexed fromFeeBP, uint16 indexed toFeeBP);
    event FeeReceiverChange(
        address indexed fromFeeReceiver,
        address indexed toFeeReceiver
    );
    event Swap(
        address indexed router,
        address indexed fromToken,
        address indexed toToken,
        uint256 amountIn,
        uint256 feesPaid
    );

    /****************
     ** Structures **
     ****************/

    struct SwapDesc {
        IUniswapV2Router01 router;
        string swapFunc;
        uint256 amountIn;
        uint256 amountOutMin;
        address[] path;
        address to;
        uint256 deadline;
        bool withFee;
    }

    /***************
     ** Modifiers **
     ***************/

    modifier pathMinLength2(address[] memory _path) {
        _pathMinLength2(_path);
        _;
    }

    /***********************
     ** Ownable Functions **
     ***********************/

    function setFeeReceiver(address _feeReceiver) external onlyOwner {
        _setFeeReceiver(_feeReceiver);
    }

    function setFeeBP(uint16 _feeBP) external onlyOwner {
        require(
            _feeBP <= FEE_DENOMINATOR,
            "UniswapV2Router01IntermediaryV2: feeBP > FEE_DENOMINATOR"
        );
        require(
            _feeBP <= MAX_FEE_BP,
            "UniswapV2Router01IntermediaryV2: feeBP > MAX_FEE_BP"
        );

        emit FeeBPChange(feeBP, _feeBP);

        feeBP = _feeBP;
    }

    /************************
     ** External Functions **
     ************************/

    function getFeeDetails(uint256 _amount)
        external
        view
        returns (uint256 _fee, uint256 _left)
    {
        return _getFeeDetails(_amount);
    }

    function swapExactTokensForTokens(
        IUniswapV2Router01 router,
        string memory swapFunc,
        uint256 amountIn,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256) {
        return
            _swapExactTokensForTokens(
                SwapDesc({
                    router: router,
                    swapFunc: swapFunc,
                    amountIn: amountIn,
                    amountOutMin: amountOutMin,
                    path: path,
                    to: to,
                    deadline: deadline,
                    withFee: false
                })
            );
    }

    function swapExactETHForTokens(
        IUniswapV2Router01 router,
        string memory swapFunc,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external payable returns (uint256) {
        return
            _swapExactETHForTokens(
                SwapDesc({
                    router: router,
                    swapFunc: swapFunc,
                    amountIn: address(this).balance,
                    amountOutMin: amountOutMin,
                    path: path,
                    to: to,
                    deadline: deadline,
                    withFee: false
                })
            );
    }

    function swapExactTokensForETH(
        IUniswapV2Router01 router,
        string memory swapFunc,
        uint256 amountIn,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256) {
        return
            _swapExactTokensForETH(
                SwapDesc({
                    router: router,
                    swapFunc: swapFunc,
                    amountIn: amountIn,
                    amountOutMin: amountOutMin,
                    path: path,
                    to: to,
                    deadline: deadline,
                    withFee: false
                })
            );
    }

    function swapExactTokensForTokensWithFee(
        IUniswapV2Router01 router,
        string memory swapFunc,
        uint256 amountIn,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256) {
        return
            _swapExactTokensForTokens(
                SwapDesc({
                    router: router,
                    swapFunc: swapFunc,
                    amountIn: amountIn,
                    amountOutMin: amountOutMin,
                    path: path,
                    to: to,
                    deadline: deadline,
                    withFee: true
                })
            );
    }

    function swapExactETHForTokensWithFee(
        IUniswapV2Router01 router,
        string memory swapFunc,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external payable returns (uint256) {
        return
            _swapExactETHForTokens(
                SwapDesc({
                    router: router,
                    swapFunc: swapFunc,
                    amountIn: address(this).balance,
                    amountOutMin: amountOutMin,
                    path: path,
                    to: to,
                    deadline: deadline,
                    withFee: true
                })
            );
    }

    function swapExactTokensForETHWithFee(
        IUniswapV2Router01 router,
        string memory swapFunc,
        uint256 amountIn,
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external returns (uint256) {
        return
            _swapExactTokensForETH(
                SwapDesc({
                    router: router,
                    swapFunc: swapFunc,
                    amountIn: amountIn,
                    amountOutMin: amountOutMin,
                    path: path,
                    to: to,
                    deadline: deadline,
                    withFee: true
                })
            );
    }

    /************************
     ** Internal Functions **
     ************************/

    function _setFeeReceiver(address _feeReceiver) internal {
        /// @dev avoid misuse by checking for zero address
        require(_feeReceiver != address(0));

        emit FeeReceiverChange(feeReceiver, _feeReceiver);

        feeReceiver = _feeReceiver;
    }

    function _getFeeDetails(uint256 _amount)
        internal
        view
        returns (uint256 _fee, uint256 _left)
    {
        _fee = (_amount * feeBP) / FEE_DENOMINATOR;
        _left = _amount - _fee;
    }

    function _pathMinLength2(address[] memory _path) internal pure {
        /// @dev path always has min. 2 tokens
        require(_path.length >= 2, "UniswapV2Router01IntermediaryV2: path < 2");
    }

    function _swapExactTokensForTokens(SwapDesc memory swapDesc)
        internal
        pathMinLength2(swapDesc.path)
        returns (uint256)
    {
        /// @dev if `path[0]` is zero the tx will fail because of SafeERC20
        IERC20 _fromToken = IERC20(swapDesc.path[0]);

        uint amountIn = swapDesc.amountIn;
        uint fee = 0;
        if (swapDesc.withFee) {
            (uint256 _fee, uint256 _left) = _getFeeDetails(swapDesc.amountIn);

            /// @dev send tokens to this contract for swapping (minus fees)
            _fromToken.safeTransferFrom(_msgSender(), address(this), _left);
            /// @dev send fees to fee receiver
            _fromToken.safeTransferFrom(_msgSender(), feeReceiver, _fee);

            /// @dev approve tx to router
            _fromToken.safeIncreaseAllowance(address(swapDesc.router), _left);

            amountIn = _left;
            fee = _fee;
        } else {
            /// @dev send tokens to this contract for swapping
            _fromToken.safeTransferFrom(
                _msgSender(),
                address(this),
                swapDesc.amountIn
            );

            /// @dev approve tx to router
            _fromToken.safeIncreaseAllowance(
                address(swapDesc.router),
                swapDesc.amountIn
            );
        }

        emit Swap(
            address(swapDesc.router),
            swapDesc.path[0],
            swapDesc.path[swapDesc.path.length - 1],
            swapDesc.amountIn,
            fee
        );

        bytes memory payload = abi.encodeWithSignature(
            string.concat(
                swapDesc.swapFunc,
                "(uint256,uint256,address[],address,uint256)"
            ),
            amountIn,
            swapDesc.amountOutMin,
            swapDesc.path,
            swapDesc.to,
            swapDesc.deadline
        );

        (bool success, bytes memory returnData) = address(swapDesc.router).call(
            payload
        );

        require(success, "UniswapV2Router01IntermediaryV2: !success");

        return uint256(bytes32(returnData));
    }

    function _swapExactETHForTokens(SwapDesc memory swapDesc)
        internal
        pathMinLength2(swapDesc.path)
        returns (uint256)
    {
        uint amountIn = swapDesc.amountIn;
        uint fee = 0;

        if (swapDesc.withFee) {
            (uint256 _fee, uint256 _left) = _getFeeDetails(swapDesc.amountIn);

            /// @dev send fees to fee receiver
            (bool sent, ) = feeReceiver.call{value: _fee}("");
            /// @dev check if Ether have been sent
            require(
                sent,
                "UniswapV2Router01IntermediaryV2: Failed to send Ether"
            );

            amountIn = _left;
            fee = _fee;
        }

        emit Swap(
            address(swapDesc.router),
            swapDesc.path[0],
            swapDesc.path[swapDesc.path.length - 1],
            swapDesc.amountIn,
            fee
        );

        bytes memory payload = abi.encodeWithSignature(
            string.concat(
                swapDesc.swapFunc,
                "(uint256,address[],address,uint256)"
            ),
            swapDesc.amountOutMin,
            swapDesc.path,
            swapDesc.to,
            swapDesc.deadline
        );

        (bool success, bytes memory returnData) = address(swapDesc.router).call{
            value: amountIn
        }(payload);

        require(success, "UniswapV2Router01IntermediaryV2: !success");

        return uint256(bytes32(returnData));
    }

    function _swapExactTokensForETH(SwapDesc memory swapDesc)
        internal
        pathMinLength2(swapDesc.path)
        returns (uint256)
    {
        /// @dev if `path[0]` is zero the tx will fail because of SafeERC20
        IERC20 _fromToken = IERC20(swapDesc.path[0]);

        uint amountIn = swapDesc.amountIn;
        uint fee = 0;

        if (swapDesc.withFee) {
            (uint256 _fee, uint256 _left) = _getFeeDetails(swapDesc.amountIn);

            /// @dev send tokens to this contract for swapping (minus fees)
            _fromToken.safeTransferFrom(_msgSender(), address(this), _left);
            /// @dev send fees to fee receiver
            _fromToken.safeTransferFrom(_msgSender(), feeReceiver, _fee);

            /// @dev approve tx to router
            _fromToken.safeIncreaseAllowance(address(swapDesc.router), _left);

            amountIn = _left;
            fee = _fee;
        } else {
            /// @dev send tokens to this contract for swapping
            _fromToken.safeTransferFrom(
                _msgSender(),
                address(this),
                swapDesc.amountIn
            );

            /// @dev approve tx to router
            _fromToken.safeIncreaseAllowance(
                address(swapDesc.router),
                swapDesc.amountIn
            );
        }

        emit Swap(
            address(swapDesc.router),
            swapDesc.path[0],
            swapDesc.path[swapDesc.path.length - 1],
            swapDesc.amountIn,
            fee
        );

        bytes memory payload = abi.encodeWithSignature(
            string.concat(
                swapDesc.swapFunc,
                "(uint256,uint256,address[],address,uint256)"
            ),
            amountIn,
            swapDesc.amountOutMin,
            swapDesc.path,
            swapDesc.to,
            swapDesc.deadline
        );

        (bool success, bytes memory returnData) = address(swapDesc.router).call(
            payload
        );

        require(success, "UniswapV2Router01IntermediaryV2: !success");

        return uint256(bytes32(returnData));
    }
}